

Minimizing the earliness–tardiness for the customer order scheduling problem in a dedicated machine environment

Matheus Peixoto Ribeiro Vieira - 22.1.4104

Pedro Henrique R. L. de Oliveira - 22.1.4022



Conjuntos e Parâmetros

```
param n; # Numero de pedidos  
param m; # Numero de maquinas
```

Sets

```
set Pedidos := {1 .. n}; # Pedidos  
set Maquinas := {1 .. m}; # Maquinas  
set Posicoes := Pedidos; # Numero de posicoes igual ao numero de pedidos
```

Parametros

```
param p{i in Pedidos, j in Maquinas}; # Tempo de processamento  
param d{i in Pedidos}; # Deadline definido previamente  
param M := 1000; # Constante grande
```

Variáveis de decisão

```
# Atribuição da tarefa j do pedido i a posição h (1 se a tarefa j do pedido i está na posição h, 0 caso  
↳ contrário)  
var x{i in Pedidos, j in Maquinas, h in Posicoes}, binary;  
# Tempo de finalização da tarefa j do pedido i quando atribuída a posição h (Conclusao_Tarefa[i,j] ↵  
↳ caso xijh = 1, 0 caso xijh = 0)  
var z{i in Pedidos, j in Maquinas, h in Posicoes} >= 0;  
# Tempo de finalização da tarefa j do pedido i  
var Conclusao_Tarefa{i in Pedidos, j in Maquinas} >= 0;  
# Tempo de finalização do pedido i  
var Conclusao_Pedido{i in Pedidos} >= 0;  
# Tempo de antecipação da tarefa j do pedido i  
var Antecipacao{i in Pedidos, j in Maquinas} >= 0;  
# Tempo de atraso do pedido  
var Atraso{i in Pedidos} >= 0;
```



Função objetivo

$$\text{minimize: } \sum_{i=1}^n \left(\sum_{j=1}^m (E_{ij}) + m \cdot T_i \right)$$

```
# Funcao objetivo
```

```
minimize Total_Cost:
```

```
    sum{i in Pedidos} (sum{j in Maquinas} Antecipacao[i,j] + m * Atraso[i]);
```

Restrições 2 e 3

$$\sum_{h=1}^n x_{ijh} = 1 \forall i \in I; j \in J$$

Garante que so tenha 1 tarefa do pedido i por maquina

```
Um_pedido_por_maquina {i in Pedidos, j in Maquinas}:  
    sum{h in Posicoes} x[i,j,h] = 1;
```

$$\sum_{i=1}^n x_{ijh} = 1 \forall h \in I; j \in J$$

Garante que so tenha 1 tarefa por posicao na maquina j

```
Uma_tarefa_por_posicao {h in Posicoes, j in Maquinas}:  
    sum{i in Pedidos} x[i,j,h] = 1;
```

Restrições 4 e 5

Garantem que uma tarefa só pode iniciar após o fim da que está na posição anterior

$$\sum_{i=1}^n z_{ij(h-1)} \leq \sum_{i=1}^n (z_{ijh} - x_{ijh} \cdot p_{ij}) \quad \forall h \in I : h \neq 1; j \in J$$

```
Restricao_de_sequencia {h in Posicoes, j in Maquinas: h > 1}:  
    sum{i in Pedidos} z[i,j,h-1] <= sum{i in Pedidos} (z[i,j,h] - x[i,j,h]*p[i,j]);
```

$$0 \leq \sum_{i=1}^n (z_{ij1} - x_{ij1} \cdot p_{ij}) \quad \forall j \in J$$

```
Inicio_primeira_tarefa {j in Maquinas}:  
    0 <= sum{i in Pedidos} (z[i,j,1] - x[i,j,1]*p[i,j]);
```

Restrições 6 e 7

Garante que $z[i,j,h]$ (tempo de finalização da tarefa j do pedido i na posição h) tenha o mesmo valor de $Conclusao_Tarefa[i,j]$ quando essa tarefa tiver sido atribuída a posição h (ou seja, quando $x[i,j,h] = 1$), e tenha valor 0 caso contrário

$$C_{ij} - M \cdot (1 - x_{ijh}) \leq z_{ijh} \quad \forall i \in I; \quad h \in I; \quad j \in J$$

```
tempo_finalizacao_z {i in Pedidos, j in Maquinas, h in Posicoes}:  
    Conclusao_Tarefa[i,j] - M * (1 - x[i,j,h]) <= z[i,j,h];
```

$$\sum_{h=1}^n z_{ijh} \leq C_{ij} \quad \forall i \in I; \quad j \in J$$

```
z_completude_unica {i in Pedidos, j in Maquinas}:  
    sum {h in Posicoes} z[i,j,h] <= Conclusao_Tarefa[i,j];
```

Restrições 8 e 9

$$C_{ij} \leq C_i \forall i \in I; j \in J$$

Garante que nenhuma tarefa do pedido i tenha um tempo de conclusao maior que o proprio pedido

```
Restricao_tempo_conclusao_tarefa{i in Pedidos, j in Maquinas}:  
    Conclusao_Tarefa[i,j] <= Conclusao_Pedido[i];
```

$$C_i - d_i \leq T_i \forall i \in I$$

Garante que o tempo de atraso do pedido i seja o valor maximo entre 0 e a subtracao do seu tempo de conclusao - seu Deadline

```
Restricao_de_atraso{i in Pedidos}:  
    Conclusao_Pedido[i] - d[i] <= Atraso[i];
```


Restrições 10 e 11

Garantem que o tempo de antecipação da tarefa j do pedido i seja a subtração do valor máximo entre o Deadline do pedido e seu tempo de conclusão - o tempo de conclusão daquela tarefa

$$d_i - C_{ij} \leq E_{ij} \quad \forall i \in I; j \in J$$


```
Restricao_de_antecipacao{i in Pedidos, j in Maquinas}:  
    d[i] - Conclusao_Tarefa[i,j] <= Antecipacao[i,j];
```

$$C_i - C_{ij} \leq E_{ij} \quad \forall i \in I; j \in J$$

```
Restricao_de_antecipacao2{i in Pedidos, j in Maquinas}:  
    Conclusao_Pedido[i] - Conclusao_Tarefa[i,j] <= Antecipacao[i,j];
```

Dados de entrada 1

```
param n := 3;  
param m := 2;  
  
param p : 1..2 :=  
1 : 2 : 1  
2 : 1 : 1  
3 : 1 : 6;  
  
param d :=  
1 : 7  
2 : 8  
3 : 9;
```



Dados de entrada 2.1


```
param n := 5;
param m := 2;

param p : 1..2 :=
  1 4 5
  2 5 6
  3 4 4
  4 6 5
  5 5 4;

param d :=
  1 18
  2 20
  3 16
  4 22
  5 19;
```


Custo: 43

Gráfico de Gantt para o Problema



Dados de entrada 2.2


```
param n := 5;  
param m := 2;  
  
param p : 1..2 :=  
..... 1..4..5  
..... 2..5..6  
..... 3..4..4  
..... 4..6..5  
..... 5..5..4;  
  
param d :=  
..... 1..7  
..... 2..20  
..... 3..15  
..... 4..25  
..... 5..11;
```



Dados de entrada 3.1

```
param n := 5;  
param m := 3;  
  
param p : 1..2..3 :=  
1 4 5 4  
2 5 6 5  
3 4 4 5  
4 6 5 4  
5 5 4 6;  
  
param d :=  
1 20  
2 22  
3 19  
4 23  
5 21;
```

Custo: 73



Dados de entrada 3.2


```
param n := 5;  
param m := 3;
```

```
param p : 1..2..3 :=  
1..4..5..4  
2..5..6..5  
3..4..4..5  
4..6..5..4  
5..5..4..6;
```

```
param d :=  
1..13  
2..6  
3..16  
4..25  
5..20;
```

Custo: 14

Gráfico de Gantt para o Problema





Conclusões

A função objetivo é muito sensível aos deadlines.

Quanto mais pedidos, tarefas por pedidos (número de máquinas) ou quanto maiores os tempos de processamento de cada tarefa, mais diferença faz ter os deadlines mais espaçados para um valor ótimo melhor.



Minimizing the earliness–tardiness for the customer order scheduling problem in a dedicated machine environment

Matheus Peixoto Ribeiro Vieira - 22.1.4104
Pedro Henrique R. L. de Oliveira - 22.1.4022